

# Testing For Linear Independence

Worksheet by Michael K. May S.J., revised by Russell Blyth

```
> restart: with(LinearAlgebra): with(plots): with(plottools):  
Warning, the name changecoords has been redefined  
Warning, the assigned name arrow now has a global binding
```

## Outline:

The basic objectives are:

- 1) Use Gauss-Jordan elimination to test the linear independence of column vectors of a matrix.
- 2) Show that a set of real valued functions is linearly independent.
- 3) Show that a set of real valued functions is linearly dependent.

```
>
```

## Part 1: The Linear Independence of the Column Vectors of a Matrix

Given a set of vectors,  $S = \{V_1, \dots, V_m\}$ , there are a number of questions we might ask concerning linear independence:

Is S a set of linearly independent vectors?

Can we find a maximal linearly independent subset of S?

Can we express the other vectors in S as linear combinations of our maximal linearly independent subset?

We first look at the case where the vectors are in  $R^n$ .

For the first example, consider a set of 5 vectors in  $R^3$

```
> v1 := Vector(3, [1, 2, 3]); v2 := Vector(3, [1, 2, 5]);  
v3 := Vector(3, [2, 4, 8]); v4 := Vector(3, [1, 1, 1]);  
v5 := Vector(3, [-4, 14, -26]);
```

$$v1 := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$v2 := \begin{bmatrix} 1 \\ 2 \\ 5 \end{bmatrix}$$

$$v3 := \begin{bmatrix} 2 \\ 4 \\ 8 \end{bmatrix}$$

$$v4 := \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$$

$$v5 := \begin{bmatrix} -4 \\ 14 \\ -26 \end{bmatrix} \quad (2.1)$$

Put the vectors together into a matrix (as column vectors).

```
> MatOfVec := <v1 | v2 | v3 | v4 | v5>;
```

$$MatOfVec := \begin{bmatrix} 1 & 1 & 2 & 1 & -4 \\ 2 & 2 & 4 & 1 & 14 \\ 3 & 5 & 8 & 1 & -26 \end{bmatrix} \quad (2.2)$$

Next we compute the rank and row reduce the matrix to find a reduced echelon form.

```
> Rank(MatOfVec);
RedMat := ReducedRowEchelonForm(MatOfVec);
```

$$RedMat := \begin{bmatrix} 1 & 0 & 1 & 0 & 47 \\ 0 & 1 & 1 & 0 & -29 \\ 0 & 0 & 0 & 1 & -22 \end{bmatrix} \quad (2.3)$$

We conclude that:

- 1) The rank of the matrix is 3, so the set of 5 column vectors cannot be linearly independent. The biggest subset of linearly independent vectors has size 3.
- 2) The pivots of the reduced echelon form correspond to independent vectors. Thus  $\{v1, v2, v4\}$  is a linearly independent set.
- 3) Reading the coefficients from the reduced matrix,  $v3$  is the sum of  $v1$  and  $v2$ , while  $v5$  is equal to 47 times  $v1$  minus 29 times  $v2$  minus 22 times  $v4$ .

To see why 3) works, compute the general solution of  $k1v1 + \dots + k5v5 = 0$ :

```
> B := Vector(3, [0, 0, 0]);
gensol := LinearSolve(MatOfVec, B);
```

$$B := \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

(2.4)

$$\text{gensol} := \begin{bmatrix} -t_3 - 47t_5 \\ -t_3 + 29t_5 \\ -t_3 \\ 22t_5 \\ -t_5 \end{bmatrix} \quad (2.4)$$

```
> eval(gensol, {_t[3]=1, _t[5]=0});
eval(gensol, {_t[3]=0, _t[5]=1});
```

$$\begin{bmatrix} -1 \\ -1 \\ 1 \\ 0 \\ 0 \end{bmatrix} \quad \begin{bmatrix} -47 \\ 29 \\ 0 \\ 22 \\ 1 \end{bmatrix}$$

(2.5)

The first particular solution says that  $v_3 = v_1 + v_2$ . The second particular solution says  $v_5 = 47v_1 - 29v_2 - 22v_4$ . Note that the reduced echelon matrix has the appropriate coefficients already computed in terms of the pivots.

```
>
```

### Exercises:

1) Explain why we are justified in using the reduced form of the matrix to draw conclusions about the original set of vectors. (Your answer should connect equations of vectors to solutions of systems of equations.)

```
>
```

2) (Read carefully:) Use the phone numbers of 5 friends to produce 7 vectors in  $\mathbb{R}^5$ . Find a maximal linearly independent subset of the vectors. Express the other vectors as linear combinations of vectors in the subset.

```
>
```

3) Is the set  $S = \left\{ \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 3 & 2 \\ -1 & 2 \end{bmatrix}, \begin{bmatrix} -2 & 1 \\ 6 & 7 \end{bmatrix}, \begin{bmatrix} -5 & -3 \\ 8 & 6 \end{bmatrix} \right\}$  of  $2 \times 2$  matrices

linearly independent? (Show the appropriate system of linear equations that you need to solve as part of your solution.)

```
>
```

□ □ >

## ▼ Part 2: Showing a set of functions is linearly independent

Another vector space we are interested in is the space of real-valued functions with domain all reals.

To show that the functions  $x^2$ ,  $\sin(x)$ , and  $\sin(2x)$  are linearly independent, we need to show that the equation

$$a_1 x^2 + a_2 \sin(x) + a_3 \sin(2x) = 0$$

has only the trivial solution (for unknowns  $a_1$ ,  $a_2$  and  $a_3$ ). Thus we need to show that the equation does not hold if any of the  $a_i$  has a nonzero value. The trick is to note that any solution that works for all values of  $x$  must work for particular values of  $x$ . Plugging in some well chosen values for  $x$  gives us linear equations in  $a_1$ ,  $a_2$ , and  $a_3$ . Since there are three unknowns we want to use at least three equations. (We will try four equations just to be safe.)

```
> veceqn := a1*x^2 + a2*sin(x) + a3*sin(2*x) = 0;  
val1 := simplify(eval(veceqn,x=Pi/4));  
val2 := simplify(eval(veceqn,x=Pi/2));  
val3 := simplify(eval(veceqn,x=Pi));  
val4 := simplify(eval(veceqn,x=3*Pi/4));
```

$$veceqn := a_1 x^2 + a_2 \sin(x) + a_3 \sin(2x) = 0$$

$$val1 := \frac{1}{16} a_1 \pi^2 + \frac{1}{2} a_2 \sqrt{2} + a_3 = 0$$

$$val2 := \frac{1}{4} a_1 \pi^2 + a_2 = 0$$

$$val3 := a_1 \pi^2 = 0$$

$$val4 := \frac{9}{16} a_1 \pi^2 + \frac{1}{2} a_2 \sqrt{2} - a_3 = 0 \tag{3.1}$$

```
> solve({val1, val2, val3, val4},{a1, a2, a3});  
{a1=0, a2=0, a3=0} \tag{3.2}
```

Since we have shown the system of equations obtained by using specific values of  $x$  only has the trivial solution, we have shown that  $veceqn$  can only have the trivial solution.

□ >

### ▼ Exercises:

□ 4) Show that the functions  $\sin(x)$ ,  $\sin(2x)$ , and  $\sin(3x)$  are linearly independent.

□ >

□ 5) Show that the functions  $e^x$ ,  $e^{2x}$  and  $e^{-x}$  are linearly independent. (Recall that the function  $e^x$  is written  $\exp(x)$  in Maple.)

□ >

□ >

## ▼ Part 3: Showing a set of functions is linearly dependent

If we show vectors are independent by showing that the vector equation has no nontrivial solutions, it follows that we show a set of vectors is dependent if we find a nontrivial solution to that equation.

As an example, suppose we want to test the independence of the set of functions  $\{e^x, \sinh(x), \cos(x), \cosh(x)\}$ . (For the time being, pretend that you have forgotten the definitions of  $\sinh(x)$  and  $\cosh(x)$ .)

```
> veceqn := a1*exp(x) + a2*sinh(x) + a3*cos(x) + a4*cosh(x) = 0;
val1 := simplify(eval(veceqn, x=0));
val2 := simplify(eval(veceqn, x=Pi/2));
val3 := simplify(eval(veceqn, x=Pi));
val4 := simplify(eval(veceqn, x=1));
val5 := simplify(eval(veceqn, x=3*Pi/2));
soll := solve({val1, val2, val3, val4, val5});
```


$$\begin{aligned} \text{veceqn} &:= a1 e^x + a2 \sinh(x) + a3 \cos(x) + a4 \cosh(x) = 0 \\ \text{val1} &:= a1 + a3 + a4 = 0 \\ \text{val2} &:= a1 e^{\left(\frac{1}{2}\pi\right)} + a2 \sinh\left(\frac{1}{2}\pi\right) + a4 \cosh\left(\frac{1}{2}\pi\right) = 0 \\ \text{val3} &:= a1 e^\pi + a2 \sinh(\pi) - a3 + a4 \cosh(\pi) = 0 \\ \text{val4} &:= a1 e + a2 \sinh(1) + a3 \cos(1) + a4 \cosh(1) = 0 \\ \text{val5} &:= a1 e^{\left(\frac{3}{2}\pi\right)} + a2 \sinh\left(\frac{3}{2}\pi\right) + a4 \cosh\left(\frac{3}{2}\pi\right) = 0 \\ \text{soll} &:= \{a3=0, a4=a2, a1=-a2, a2=a2\} \end{aligned} \tag{4.1}$$

Note that instead of finding only the trivial solution, we get a one parameter family of solutions. Setting  $a4$  to 1, we get the equation

$$-(e^x) + \sinh(x) + \cosh(x) = 0$$

which is a dependency relation if it is true. (It follows easily from the definition of  $\sinh(x)$  and  $\cosh(x)$ .) Now that we have a candidate nontrivial solution for the vector equation, we want to see if it works for all values of  $x$ , or perhaps only for the 5 values we chose. We can check intervals by plotting the graph.

```
> plot({-exp(x) + sinh(x) + cosh(x), x}, x=-10..10, axes=boxed);
```



(We plotted the function  $y=x$  as well to give a reasonable viewing window. If we plot just the linear combination we can get problems arising from cancellation errors.) Thus we see that we have found a dependency relationship between the functions, so they are linearly dependent. (Actually, we only showed the functions are dependent on the domain from -10 to 10. We rely on a theorem beyond the scope of this class to claim that we have shown enough.)

The technique we used to find the coefficients of a dependence relationship can run into trouble if the values of  $x$  we choose fit too nicely. Consider a slight variation on the functions from exercise 4 above

```
> veceqn := a1*sin(x) + a2*sin(2*x) + a3*sin(4*x) = 0;
   val1 := simplify(eval(veceqn, x=0));
   val2 := simplify(eval(veceqn, x=Pi/2));
   val3 := simplify(eval(veceqn, x=Pi/4));
   val4 := simplify(eval(veceqn, x=3*Pi/4));
   val5 := simplify(eval(veceqn, x=Pi));
   sol1 := solve({val1, val2, val3, val4, val5}, {a1, a2, a3});
           veceqn := a1 sin(x) + a2 sin(2x) + a3 sin(4x) = 0
                   val1 := 0 = 0
                   val2 := a1 = 0
```

$$val3 := \frac{1}{2} a1 \sqrt{2} + a2 = 0$$

$$val4 := \frac{1}{2} a1 \sqrt{2} - a2 = 0$$

$$val5 := 0 = 0$$

$$soll := \{a3 = a3, a1 = 0, a2 = 0\} \quad (4.2)$$

Since  $a3$  is a parameter which does not appear elsewhere in the solution, this seems to suggest that  $\sin(4x) = 0$ . Clearly this is false. We need to include a point where the third function is not zero.

>

### Exercises:

6) Find a dependence relationship for the functions  $e^x$ ,  $e^{-x}$ , and  $\sinh(x)$ .

>

7) Find a dependence relationship for the functions  $e^x$ ,  $e^{-x}$ , and  $\cosh(x)$ .

>

8) As mentioned above, the technique we used to find the coefficients of a dependence relationship can run into trouble if the values of  $x$  we choose fit too nicely. Explore what happens if we test the linear independence of  $\sin(x)$ ,  $\sin(3x)$ , and  $\sin(5x)$  with  $x$  taking on the values  $0$ ,  $\pi/2$ ,  $\pi$ ,  $3\pi/2$ , and  $2\pi$ . Graphically show that the proposed dependence relation is false. Show that the functions are actually linearly independent.

>

>

>